Determine orientation by correlation

Workspace describing a method to extract a vector measurement from directional measurements by means of forming the cross correlation

■ Initialization

```
In[1]:= Off[General::spell, General::spell1]

In[2]:= LSolve = Last[Solve[##]] &

Out[2]= Last[Solve[##1]] &

In[3]:= Needs["Calculus`VectorAnalysis`"]
        Needs["Statistics`MultiDescriptiveStatistics`"]
        Needs["Statistics`ContinuousDistributions`"]

In[6]:= PeS = PowerExpand[Simplify[#]] &
        FpPeS = FixedPoint[PeS, #] &

Out[6]= PowerExpand[Simplify[#1]] &

Out[7]= FixedPoint[PeS, #1] &
```

■ Geometric sum (magnitude of a vector in Euclidean metrics)

```
In[8]:= GeomSum = Function[{v}, √(Plus @@ (#² & /@ v))]

Out[8]= Function[{v}, √(Plus @@ (#1² &) /@ v)]
```

■ Basic principle: Reconstructing a vector from its autocorrelation

We define a 3 component vector,
in this particular case in sperical coordinates

```
In[9]:= Bvec = CoordinatesToCartesian[{B, θ, ϕ}, Spherical]

Out[9]= {B Cos[ϕ] Sin[θ], B Sin[θ] Sin[ϕ], B Cos[θ]}
```

The correlation of the vector is obtained by forming its outer product with itself

```
In[10]:= MatrixForm[Bcor = Outer[Times, Bvec, Bvec]]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} B^2 \mathrm{Cos}[\phi]^2 \mathrm{Sin}[\theta]^2 & B^2 \mathrm{Cos}[\phi] \mathrm{Sin}[\theta]^2 \mathrm{Sin}[\phi] & B^2 \mathrm{Cos}[\theta] \mathrm{Cos}[\phi] \mathrm{Sin}[\theta] \\ B^2 \mathrm{Cos}[\phi] \mathrm{Sin}[\theta]^2 \mathrm{Sin}[\phi] & B^2 \mathrm{Sin}[\theta]^2 \mathrm{Sin}[\phi]^2 & B^2 \mathrm{Cos}[\theta] \mathrm{Sin}[\theta] \mathrm{Sin}[\phi] \\ B^2 \mathrm{Cos}[\theta] \mathrm{Cos}[\phi] \mathrm{Sin}[\theta] & B^2 \mathrm{Cos}[\theta] \mathrm{Sin}[\theta] \mathrm{Sin}[\phi] & B^2 \mathrm{Cos}[\theta]^2 \end{pmatrix}$$

The eigensystem of the correlation matrix yields

```
In[11]:= Simplify[Eigensystem[Bcor]]

Out[11]= {{0, 0, B²}, {{-Cot[θ] Sec[ϕ], 0, 1}, {-Tan[ϕ], 1, 0}, {Cos[ϕ] Tan[θ], Sin[ϕ] Tan[θ], 1}}}
```

two eigenvectors to the eigenvalue 0,
and one non-zero eigenvector in the direction of the original vector, with the eigenvalue being the square of the magnitude of the vector.

```
In[12]:= TableForm[Transpose[{Eigenvalues[Bcor], MatrixForm /@ Eigenvectors[Bcor]}]]
```

Out[12]//TableForm=

$$0 \quad \begin{pmatrix} -\mathrm{Cot}[\theta] \mathrm{Sec}[\phi] \\ 0 \\ 1 \end{pmatrix}$$

$$0 \quad \begin{pmatrix} -\mathrm{Tan}[\phi] \\ 1 \\ 0 \end{pmatrix}$$

$$B^2 \quad \begin{pmatrix} -\mathrm{Cot}[\theta] \mathrm{Sec}[\phi] + \mathrm{Csc}[\theta] \mathrm{Sec}[\theta] \mathrm{Sec}[\phi] - \mathrm{Sin}[\phi] \mathrm{Tan}[\theta] \mathrm{Tan}[\phi] \\ \mathrm{Sin}[\phi] \mathrm{Tan}[\theta] \\ 1 \end{pmatrix}$$

The direction of the eigenvector becomes more apparent as we multiply the last eigenvector by Cos[θ] and compare the result to the original vector

```
In[13]:= TableForm[MatrixForm /@ {Simplify[Last[Eigenvectors[Bcor]] Cos[θ]], Bvec}]
```

Out[13]//TableForm=

$$\begin{pmatrix} \mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta] \\ \mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi] \\ \mathrm{Cos}[\theta] \end{pmatrix}$$

$$\begin{pmatrix} \mathrm{B}\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta] \\ \mathrm{B}\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi] \\ \mathrm{B}\,\mathrm{Cos}[\theta] \end{pmatrix}$$

As an aside, the trace of the correlation matrix also yields the square of the magnitude

```
In[14]:= Tr[Bcor]
         Simplify[%]
```

Out[14]= $B^2\,\mathrm{Cos}[\theta]^2 + B^2\,\mathrm{Cos}[\phi]^2\,\mathrm{Sin}[\theta]^2 + B^2\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi]^2$

Out[15]= B^2

■ Application: Signal with random amplitude, but fixed orientation

First, we generate a "sufficiently long" scalar random sequence,
in this particular case uniformly distributed in the interval [-1,1]

```
In[16]:= rvec = Table[Random[Real, {-1, 1}], {n, 4096}];
```

and generate a sequence of vectors oriented along a given direction

```
In[17]:= rndcorrel = Transpose[rvec # & /@ Bvec];
```

The covariance matrix of the components of the random sequence is, unsurprisingly, the matrix obtained above,
times the variance of the random sequence

```
In[18]:= CovarianceMatrix[rndcorrel] // MatrixForm
```

Out[18]//MatrixForm=

$$\begin{pmatrix} 0.333575\,B^2\,\mathrm{Cos}[\phi]^2\,\mathrm{Sin}[\theta]^2 & 0.333575\,B^2\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi] & 0.333575\,B^2\,\mathrm{Cos}[\theta]\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta] \\ 0.333575\,B^2\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi] & 0.333575\,B^2\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi]^2 & 0.333575\,B^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi] \\ 0.333575\,B^2\,\mathrm{Cos}[\theta]\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta] & 0.333575\,B^2\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi] & 0.333575\,B^2\,\mathrm{Cos}[\theta]^2 \end{pmatrix}$$

as we verify by calculating the variance as $\langle x^2 \rangle - \langle x \rangle^2$

In[19]:=

$$\frac{\{\mathrm{Integrate}[\rho, \{\rho, -1, 1\}],\ \mathrm{Integrate}[\rho^2, \{\rho, -1, 1\}]\}}{\mathrm{Integrate}[1, \{\rho, -1, 1\}]}$$

$$\%[[2]] - \%[[1]]^2$$

Out[19]= $\left\{0, \frac{1}{3}\right\}$

Out[20]= $\frac{1}{3}$

Diagonalization of the random sequence (with numerical values for $\{B, \theta, \phi\}$ because even *Mathematica* will hiccup with the symbolic solution) allows us to recover the vector orientation of the random sequence

```
In[21]:= Eigensystem[CovarianceMatrix[rndcorrel /. {B → 1, θ → .5, ϕ → .6}]]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[CovarianceMatrix[rndcorrel /. {B → 1, θ → .5, ϕ → .6}]]]]]
```

Out[21]= $\{\{0.333575, -2.13533\times10^{-16}, 3.87868\times10^{-17}\},$
$\{\{0.395687, 0.270704, 0.877583\}, \{-0.495299, -0.741788, 0.452138\}, \{0.798418, -0.573602, -0.183057\}\}\}$

Out[22]//TableForm=

0.333575	$\begin{pmatrix} 0.395687 \\ 0.270704 \\ 0.877583 \end{pmatrix}$
0	$\begin{pmatrix} -0.495299 \\ -0.741788 \\ 0.452138 \end{pmatrix}$
0	$\begin{pmatrix} 0.798418 \\ -0.573602 \\ -0.183057 \end{pmatrix}$

as we can verify by comparing it to the eigensystem of the correlation of the orientation vector

```
In[23]:= Eigensystem[Bcor /. {B → 1, θ → .5, ϕ → .6}]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[Bcor /. {B → 1, θ → .5, ϕ → .6}]]]]
```

Out[23]= $\{\{1., -2.40925\times10^{-17}, 0.\}, \{\{0.395687, 0.270704, 0.877583\}, \{-0.81693, -0.332833, 0.471007\}, \{-0.918385, 0.116633, 0.378107\}\}\}$

Out[24]//TableForm=

$$\begin{array}{ll} 1. & \begin{pmatrix} 0.395687 \\ 0.270704 \\ 0.877583 \end{pmatrix} \\ 0 & \begin{pmatrix} -0.81693 \\ -0.332833 \\ 0.471007 \end{pmatrix} \\ 0 & \begin{pmatrix} -0.918385 \\ 0.116633 \\ 0.378107 \end{pmatrix} \end{array}$$

■ Application to uncorrelated random noise

■ Unbiased noise

First, we generate a sequence of random vectors,
each component of which is uniformly distributed in the interval [-1,1] in this particular case

```
In[25]:= rvec3 = Table[Table[Random[Real, {-1, 1}], {m, 3}], {n, 4096}];
```

The covariance matrix is essentially diagonal, with the diagonal elements essentially equal to the variance of the individual distributions

```
In[26]:= CovarianceMatrix[rvec3] // MatrixForm
```

Out[26]//MatrixForm=

$$\begin{pmatrix} 0.337174 & 0.00219557 & -0.00114773 \\ 0.00219557 & 0.337327 & -0.000386452 \\ -0.00114773 & -0.000386452 & 0.33796 \end{pmatrix}$$

Consequently, the eigenvalues are essentially equal

```
In[27]:= Eigensystem[CovarianceMatrix[rvec3]]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[CovarianceMatrix[rvec3]]]]]
```

Out[27]= {{0.340025, 0.337495, 0.334941}, {{-0.649258, -0.596053, 0.472424}, {0.18762, 0.476433, 0.858959}, {0.737063, -0.646322, 0.197497}}}

Out[28]//TableForm=

$$\begin{array}{ll} 0.340025 & \begin{pmatrix} -0.649258 \\ -0.596053 \\ 0.472424 \end{pmatrix} \\ 0.337495 & \begin{pmatrix} 0.18762 \\ 0.476433 \\ 0.858959 \end{pmatrix} \\ 0.334941 & \begin{pmatrix} 0.737063 \\ -0.646322 \\ 0.197497 \end{pmatrix} \end{array}$$

Obviously, the trace of the covariance matrix is the sum of the noise variances

```
In[29]:= Tr[CovarianceMatrix[rvec3]]
```

Out[29]= 1.01246

■ Biased random noise

Now we generate biased random noise

```
In[30]:= rvec3b = Table[Table[Random[Real, {0, 1}], {m, 3}], {n, 4096}];
```

Mean, second moment
variance
of the random sequence.

In[31]:= $\dfrac{\{\text{Integrate}[\rho, \{\rho, 0, 1\}], \text{Integrate}[\rho^2, \{\rho, 0, 1\}]\}}{\text{Integrate}[1, \{\rho, 0, 1\}]}$

$\%[[2]] - \%[[1]]^2$

Out[31]= $\left\{\frac{1}{2}, \frac{1}{3}\right\}$

Out[32]= $\frac{1}{12}$

Again, the covariance matrix is essentially diagonal

```
In[33]:= CovarianceMatrix[rvec3b] // MatrixForm

Out[33]//MatrixForm=
  ( 0.0846759    0.000341205  0.00106237  )
  ( 0.000341205  0.0845054    0.000409987 )
  ( 0.00106237   0.000409987  0.0838841   )
```

Since we're calculating the covariance (not the correlation), the bias is removed and we still get eigenvalues which are essentially equal to the variance of the component random sequences.

```
In[34]:= Eigensystem[CovarianceMatrix[rvec3b]]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[CovarianceMatrix[rvec3b]]]]]

Out[34]= {{0.0856469, 0.0842882, 0.0831302},
          {{-0.734411, -0.412947, -0.538623}, {-0.403246, 0.903831, -0.143116}, {-0.545924, -0.112091, 0.830303}}}

Out[35]//TableForm=
  0.0856469     (-0.734411)
                (-0.412947)
                (-0.538623)
  0.0842882     (-0.403246)
                ( 0.903831)
                (-0.143116)
  0.0831302     (-0.545924)
                (-0.112091)
                ( 0.830303)

In[36]:= N[1/12]

Out[36]= 0.0833333
```

However, the bias does show up in the correlation matrix.

While nothing changes between correlation and covariance matrix for zero mean random noise

```
In[37]:= MatrixForm[(Plus @@ # / (Length[#] - 1)) &[Outer[Times, #, #] & /@ rvec3]]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[%]]]]

Out[37]//MatrixForm=
  ( 0.337223     0.00220385   -0.00121838 )
  ( 0.00220385   0.337329     -0.00039834 )
  ( -0.00121838  -0.00039834  0.338062    )

Out[38]//TableForm=
  0.340127     (-0.647607)
               (-0.580259)
               ( 0.493867)
  0.337539     ( 0.200141)
               ( 0.495848)
               ( 0.845032)
  0.334947     ( 0.73522 )
               (-0.646091)
               ( 0.204981)
```

the bias of the random sequences sticks out like a signal:
If all the variances are equal, one eigenvector is equal to $(\sum \langle x_n \rangle^2) + \langle x^2 \rangle$, the other ones are $\langle x^2 \rangle$

```
In[39]:= MatrixForm[(Plus @@ # / (Length[#] - 1)) &[Outer[Times, #, #] & /@ rvec3b]]
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[%]]]]

Out[39]//MatrixForm=
  ( 0.33982   0.251835  0.253439 )
  ( 0.251835  0.3324    0.249176 )
  ( 0.253439  0.249176  0.333524 )

Out[40]//TableForm=
  0.838247      (-0.582603)
                (-0.573657)
                (-0.575752)
  0.0843443     (-0.556305)
                ( 0.797911)
                (-0.232083)
  0.0831523     ( 0.592536)
                ( 0.185081)
                (-0.783994)
```

$(\sum \langle x_n \rangle^2) + \langle x^2 \rangle$

In[41]:= $\frac{3}{4}+\frac{1}{12}$
```
N[%]
```

Out[41]= $\frac{5}{6}$

Out[42]= 0.833333

Another case for which the eigenvectors of the correlation matrix are predictable is the case of uncorrelated, zero mean, orthogonal, noise sources with different variances

In[43]:=
```
MatrixForm[((Plus @@ #)/(Length[#] - 1)) &[Outer[Times, #, #] & /@ Table[Table[Random[Real, {-m, m}], {m, 3}], {n, 16384}]]]
TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[%]]]]
```

Out[43]//MatrixForm=

$$\begin{pmatrix} 0.331944 & 0.00181022 & 0.00208487 \\ 0.00181022 & 1.33737 & -0.0207789 \\ 0.00208487 & -0.0207789 & 3.00055 \end{pmatrix}$$

Out[44]//TableForm=

$$\begin{array}{ll} 3.00081 & \begin{pmatrix} -0.00077265 \\ 0.0124897 \\ -0.999922 \end{pmatrix} \\ 1.33711 & \begin{pmatrix} 0.00182668 \\ 0.99992 \\ 0.0124882 \end{pmatrix} \\ 0.331939 & \begin{pmatrix} 0.999998 \\ -0.00181689 \\ -0.000795403 \end{pmatrix} \end{array}$$

In[45]:=
```
Table[Variance[UniformDistribution[-m, m]], {m, 3}]
```

Out[45]= $\left\{\frac{1}{3}, \frac{4}{3}, 3\right\}$

■ Application to a noisy measurement

We add zero mean, isotropic noise to the oriented random signal

In[46]:=
```
CovarianceMatrix[(rndcorrel /. {B → 1, θ → .5, ϕ → .6}) + rvec3] // MatrixForm
TableForm[
 Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[CovarianceMatrix[(rndcorrel /. {B → 1, θ → .5, ϕ → .6}) + rvec3]]]]]
```

Out[46]//MatrixForm=

$$\begin{pmatrix} 0.383324 & 0.0376407 & 0.105282 \\ 0.0376407 & 0.364225 & 0.0810138 \\ 0.105282 & 0.0810138 & 0.583045 \end{pmatrix}$$

Out[47]//TableForm=

$$\begin{array}{ll} 0.655654 & \begin{pmatrix} -0.379727 \\ -0.292974 \\ -0.877481 \end{pmatrix} \\ 0.340022 & \begin{pmatrix} 0.650318 \\ 0.590072 \\ -0.478436 \end{pmatrix} \\ 0.334918 & \begin{pmatrix} 0.657947 \\ -0.752317 \\ -0.0335396 \end{pmatrix} \end{array}$$

As the noise increases, the eigenvector for the largest eigenvalue deviates increasingly from the direction of the oriented signal.
Note the sign ambiguity of the eigenvector.

```
In[48]:= Table[TableForm[Transpose[
           Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[CovarianceMatrix[(rndcorrel /. {B → 1, θ → .5, ϕ → .6}) + ν rvec3]]]]], {ν, 0, 4, .5}]
```

Out[48]= {

0.333575	$\begin{pmatrix}0.395687\\0.270704\\0.877583\end{pmatrix}$	0.410229	$\begin{pmatrix}-0.388227\\-0.281242\\-0.8776\end{pmatrix}$	0.655654	$\begin{pmatrix}-0.379727\\-0.292974\\-0.877481\end{pmatrix}$
0	$\begin{pmatrix}-0.495299\\-0.741788\\0.452138\end{pmatrix}$	0.0850056	$\begin{pmatrix}0.650666\\0.590732\\-0.477147\end{pmatrix}$	0.340022	$\begin{pmatrix}0.650318\\0.590072\\-0.478436\end{pmatrix}$
0	$\begin{pmatrix}0.798418\\-0.573602\\-0.183057\end{pmatrix}$	0.0837353	$\begin{pmatrix}0.652621\\-0.756266\\-0.046344\end{pmatrix}$	0.334918	$\begin{pmatrix}0.657947\\-0.752317\\-0.0335396\end{pmatrix}$

,

1.06989	$\begin{pmatrix}0.370237\\0.30583\\0.87715\end{pmatrix}$	1.65297	$\begin{pmatrix}0.359849\\0.319699\\0.876528\end{pmatrix}$	2.40493	$\begin{pmatrix}-0.348696\\-0.334439\\-0.875535\end{pmatrix}$
0.76505	$\begin{pmatrix}0.649953\\0.589348\\-0.479823\end{pmatrix}$	1.36009	$\begin{pmatrix}0.649565\\0.588544\\-0.481333\end{pmatrix}$	2.12514	$\begin{pmatrix}-0.649144\\-0.587641\\0.483001\end{pmatrix}$
0.753513	$\begin{pmatrix}0.663691\\-0.747755\\-0.0194233\end{pmatrix}$	1.33948	$\begin{pmatrix}-0.669757\\0.742569\\0.00412161\end{pmatrix}$	2.09279	$\begin{pmatrix}0.676035\\-0.736769\\0.0121912\end{pmatrix}$

,

3.3258	$\begin{pmatrix}-0.336958\\-0.349881\\-0.874095\end{pmatrix}$	4.41562	$\begin{pmatrix}0.324855\\0.365842\\0.87214\end{pmatrix}$	5.67441	$\begin{pmatrix}-0.312651\\-0.382143\\-0.869607\end{pmatrix}$
3.0602	$\begin{pmatrix}0.648681\\0.586613\\-0.48487\end{pmatrix}$	4.16526	$\begin{pmatrix}-0.648161\\-0.585423\\0.486998\end{pmatrix}$	5.44034	$\begin{pmatrix}-0.647565\\-0.584024\\0.489465\end{pmatrix}$
3.01341	$\begin{pmatrix}0.682402\\-0.73039\\0.0292977\end{pmatrix}$	4.1013	$\begin{pmatrix}0.688735\\-0.723491\\0.0469467\end{pmatrix}$	5.35645	$\begin{pmatrix}-0.694917\\0.716158\\-0.0648663\end{pmatrix}$

}

The deviation of the noisy measurement from the noiseless measurement can be obtained by forming the cross product of the noiseless and the noisy eigenvector. This formulation exploits the feature of *Mathematica* to normalize the eigenvectors to unit length.

```
In[49]:= Table[First[Eigenvectors[CovarianceMatrix[(rndcorrel /. {B → 1, θ → .5, ϕ → .6}) + ν rvec3]]], {ν, 0, 20, .5}]
         MatrixForm /@ (aux = (Cross[#, First[%]] & /@ %))
```

```
Out[49]= {{0.395687, 0.270704, 0.877583}, {-0.388227, -0.281242, -0.8776}, {-0.379727, -0.292974, -0.877481}, {0.370237, 0.30583, 0.87715},
  {0.359849, 0.319699, 0.876528}, {-0.348696, -0.334439, -0.875535}, {-0.336958, -0.349881, -0.874095}, {0.324855, 0.365842, 0.87214},
  {-0.312651, -0.382143, -0.869607}, {0.300643, 0.398629, 0.866434}, {0.28917, 0.415199, 0.862549}, {-0.278622, -0.431836, -0.857839},
  {-0.269482, -0.448657, -0.852107}, {-0.262419, -0.466008, -0.844969}, {-0.258522, -0.484652, -0.835631},
  {-0.259905, -0.506271, -0.822277}, {0.271661, 0.535, 0.799985}, {-0.310027, -0.583442, -0.750652}, {-0.448291, -0.697191, -0.559429},
  {0.662666, 0.746682, -0.0577877}, {0.672546, 0.681163, -0.289306}, {-0.667789, -0.652948, 0.357375}, {-0.66436, -0.638728, 0.388139},
  {-0.662045, -0.630278, 0.405519}, {-0.660408, -0.624696, 0.416674}, {-0.659193, -0.620736, 0.424442},
  {-0.658256, -0.617781, 0.430169}, {-0.657509, -0.615491, 0.434572}, {-0.6569, -0.613662, 0.438065}, {-0.656393, -0.612167, 0.440908},
  {-0.655962, -0.610922, 0.44327}, {-0.655592, -0.609867, 0.445266}, {-0.655269, -0.608963, 0.446976}, {-0.654985, -0.608178, 0.448459},
  {-0.654732, -0.607489, 0.449758}, {-0.654506, -0.606881, 0.450908}, {-0.654302, -0.606339, 0.451932}, {-0.654117, -0.605852, 0.452852},
  {-0.653947, -0.605413, 0.453682}, {-0.653792, -0.605015, 0.454436}, {-0.653649, -0.604652, 0.455125}}
```

Out[50]=
$$\left\{\begin{pmatrix}2.21968\times10^{-17}\\3.56566\times10^{-18}\\3.16519\times10^{-18}\end{pmatrix}, \begin{pmatrix}-0.00924296\\-0.0065536\\0.00618905\end{pmatrix}, \begin{pmatrix}-0.0195714\\-0.0139664\\0.0131326\end{pmatrix}, \begin{pmatrix}0.0309427\\0.0221636\\-0.0207882\end{pmatrix}, \begin{pmatrix}0.0432826\\0.031034\\-0.0290883\end{pmatrix}, \begin{pmatrix}-0.0564873\\-0.0404283\\0.0379399\end{pmatrix}, \begin{pmatrix}-0.0704286\\-0.05016\\0.0472277\end{pmatrix}, \begin{pmatrix}0.0849648\\0.0600071\\-0.0568193\end{pmatrix},\right.$$
$$\begin{pmatrix}-0.0999555\\-0.0697152\\0.066573\end{pmatrix}, \begin{pmatrix}0.115283\\0.0789977\\-0.076347\end{pmatrix}, \begin{pmatrix}0.130876\\0.087529\\-0.0860094\end{pmatrix}, \begin{pmatrix}-0.146751\\-0.0949221\\0.0954477\end{pmatrix}, \begin{pmatrix}-0.163065\\-0.100675\\0.104578\end{pmatrix}, \begin{pmatrix}-0.180224\\-0.104049\\0.113355\end{pmatrix}, \begin{pmatrix}-0.199113\\-0.103774\\0.121787\end{pmatrix}, \begin{pmatrix}-0.221701\\-0.0972758\\0.129967\end{pmatrix},$$
$$\begin{pmatrix}0.252947\\0.0781389\\-0.138153\end{pmatrix}, \begin{pmatrix}-0.308814\\-0.0249487\\0.146935\end{pmatrix}, \begin{pmatrix}-0.460403\\0.172053\\0.154515\end{pmatrix}, \begin{pmatrix}0.670919\\-0.60441\\-0.116066\end{pmatrix}, \begin{pmatrix}0.676094\\-0.704689\\-0.0874666\end{pmatrix}, \begin{pmatrix}-0.669758\\0.727449\\0.0775896\end{pmatrix}, \begin{pmatrix}-0.665608\\0.736612\\0.0728916\end{pmatrix}, \begin{pmatrix}-0.662897\\0.741458\\0.0701744\end{pmatrix},$$
$$\begin{pmatrix}-0.661017\\0.744435\\0.0684089\end{pmatrix}, \begin{pmatrix}-0.659646\\0.746442\\0.0671711\end{pmatrix}, \begin{pmatrix}-0.658603\\0.747886\\0.0662557\end{pmatrix}, \begin{pmatrix}-0.657784\\0.748973\\0.0655512\end{pmatrix}, \begin{pmatrix}-0.657125\\0.749821\\0.0649925\end{pmatrix}, \begin{pmatrix}-0.656583\\0.7505\\0.0645384\end{pmatrix}, \begin{pmatrix}-0.656129\\0.751057\\0.0641622\end{pmatrix}, \begin{pmatrix}-0.655744\\0.751522\\0.0638453\end{pmatrix}, \begin{pmatrix}-0.655413\\0.751915\\0.0635747\end{pmatrix},$$
$$\left.\begin{pmatrix}-0.655126\\0.752252\\0.063341\end{pmatrix}, \begin{pmatrix}-0.654873\\0.752545\\0.063137\end{pmatrix}, \begin{pmatrix}-0.654651\\0.752801\\0.0629574\end{pmatrix}, \begin{pmatrix}-0.654452\\0.753027\\0.0627982\end{pmatrix}, \begin{pmatrix}-0.654274\\0.753229\\0.0626559\end{pmatrix}, \begin{pmatrix}-0.654114\\0.753409\\0.062528\end{pmatrix}, \begin{pmatrix}-0.653969\\0.753571\\0.0624125\end{pmatrix}, \begin{pmatrix}-0.653836\\0.753718\\0.0623076\end{pmatrix}\right\}$$

The magnitude of the cross product is a good indicator for the direction error

```
In[51]:= GeomSum /@ aux
```

```
Out[51]= {2.27031×10^-17, 0.0129107, 0.0273965, 0.0433685, 0.0606846, 0.0791498, 0.0985224, 0.118526,
  0.138864, 0.159247, 0.179409, 0.199139, 0.218317, 0.236973, 0.255436, 0.274782, 0.298621, 0.342897, 0.515217,
  0.910448, 0.98048, 0.991856, 0.995462, 0.997054, 0.997901, 0.998409, 0.99874, 0.998968, 0.999134, 0.999258,
  0.999354, 0.99943, 0.999492, 0.999543, 0.999585, 0.99962, 0.999651, 0.999677, 0.9997, 0.99972, 0.999737}
```

```
In[52]:= ListPlot[Transpose[{Range[0, 20, .5], GeomSum /@ aux}], Frame → True];
```



■ Sensitivity to systematic errors

■ Gain mismatch

Perturb the input signal by different gain factors aξ, aυ, aζ.
Show the eigensystem of the correlation matrix of the perturbed vector
Express the error vector as difference of the perturbed and unperturbed signal vectors
Calculate the magnitude of the error vector
Give an upper limit for the magnitude of the error as function of the maximum deviation ϵ of the gain factors aξ, aυ, aζ from unity.

```
In[53]:= Bvec {aξ, aυ, aζ}
         Simplify[Outer[Times, %, %]];
         TableForm[Transpose[Chop[{Eigenvalues[#], MatrixForm /@ Eigenvectors[#]} &[%]]]]
         Last[Last[Last[%]]] Cos[θ] aζ
         Simplify[% - (% /. {aξ → 1, aυ → 1, aζ → 1})]
         FpPeS[GeomSum[%]]
         FpPeS[% /. {aξ → 1 + ϵ, aυ → 1 + ϵ, aζ → 1 + ϵ}]
```

Out[53]= $\{a\xi\, B \cos[\phi] \sin[\theta],\ a\upsilon\, B \sin[\theta] \sin[\phi],\ a\zeta\, B \cos[\theta]\}$

Out[55]//TableForm=

0	$\begin{pmatrix} -\frac{a\zeta \cot[\theta] \sec[\phi]}{a\xi} \\ 0 \\ 1 \end{pmatrix}$
0	$\begin{pmatrix} -\frac{a\upsilon \tan[\phi]}{a\xi} \\ 1 \\ 0 \end{pmatrix}$
$a\zeta^2 B^2 \cos[\theta]^2 + a\xi^2 B^2 \cos[\phi]^2 \sin[\theta]^2 + a\upsilon^2 B^2 \sin[\theta]^2 \sin[\phi]^2$	$\begin{pmatrix} \frac{a\xi \cos[\phi] \tan[\theta]}{a\zeta} \\ \frac{a\upsilon \sin[\phi] \tan[\theta]}{a\zeta} \\ 1 \end{pmatrix}$

Out[56]= $\{a\xi \cos[\phi] \sin[\theta],\ a\upsilon \sin[\theta] \sin[\phi],\ a\zeta \cos[\theta]\}$

Out[57]= $\{(-1 + a\xi) \cos[\phi] \sin[\theta],\ (-1 + a\upsilon) \sin[\theta] \sin[\phi],\ (-1 + a\zeta) \cos[\theta]\}$

Out[58]= $\sqrt{(-1 + a\zeta)^2 \cos[\theta]^2 + (-1 + a\xi)^2 \cos[\phi]^2 \sin[\theta]^2 + (-1 + a\upsilon)^2 \sin[\theta]^2 \sin[\phi]^2}$

Out[59]= ϵ

■ Phase mismatch for a sinusoidal signal

In cartesian coordinates, because spherical polar coordinates get too messy to solve

```
In[60]:= √2 {x, y, z} Sin[2 π t + {δξ, δυ, δζ}]
         MatrixForm[CartCor = Simplify[Outer[Integrate[#1 #2, {t, 0, 1}] &, %, %]]]
```

Out[60]= $\{\sqrt{2}\, x \sin[2\pi t + \delta\xi],\ \sqrt{2}\, y \sin[2\pi t + \delta\upsilon],\ \sqrt{2}\, z \sin[2\pi t + \delta\zeta]\}$

Out[61]//MatrixForm=

$$\begin{pmatrix} x^2 & x y \cos[\delta\xi - \delta\upsilon] & x z \cos[\delta\zeta - \delta\xi] \\ x y \cos[\delta\xi - \delta\upsilon] & y^2 & y z \cos[\delta\zeta - \delta\upsilon] \\ x z \cos[\delta\zeta - \delta\xi] & y z \cos[\delta\zeta - \delta\upsilon] & z^2 \end{pmatrix}$$

```
In[62]:= eigsys = Eigensystem[CartCor];

In[63]:= Chop[eigsys /. {x → 1, y → 1, z → 1} /. Last[Solve[{δξ - δυ == ξυ, δζ - δξ == ζξ, δζ - δυ == υζ}, {ξυ, ζξ, υζ}]] /. {δξ → π/6., δυ → 0, δζ → -π/6.}]
```

Out[63]= {{2.5, 0, 0.5}, {{1., 1.1547, 1}, {1., -1.73205, 1}, {-1., 0, 1}}}

```
In[64]:= Chop[Eigensystem[
    CartCor /. {x → 5, y → 2, z → 3} /. Last[Solve[{δξ - δυ == ξυ, δζ - δξ == ζξ, δζ - δυ == υζ}, {ξυ, ζξ, υζ}]] /. {δξ → π/12., δυ → 0, δζ → -π/8.}]]

Out[64]= {{35.2987, 2.70134, 0}, {{-0.829847, -0.335121, -0.446148}, {-0.506028, 0.115045, 0.85481}, {0.235138, -0.935125, 0.26505}}}

In[65]:= Chop[
    Eigensystem[CartCor /. {x → 5, y → 2, z → 3} /. Last[Solve[{δξ - δυ == ξυ, δζ - δξ == ζξ, δζ - δυ == υζ}, {ξυ, ζξ, υζ}]] /. {δξ → 0., δυ → 0., δζ → 0.}]]

Out[65]= {{38., 0, 0}, {{0.811107, 0.324443, 0.486664}, {0.584898, -0.449921, -0.674882}, {0.246628, -0.944179, 0.218406}}}

In[66]:= CartCor /. Cos[α_] → Normal[Series[Cos[α], {α, 0, 3}]]
```

Out[66]=

$$\left\{\left\{x^2,\ x\,y\left(1-\frac{1}{2}(\delta\xi-\delta\upsilon)^2\right),\ x\,z\left(1-\frac{1}{2}(\delta\zeta-\delta\xi)^2\right)\right\},\right.$$
$$\left\{x\,y\left(1-\frac{1}{2}(\delta\xi-\delta\upsilon)^2\right),\ y^2,\ y\,z\left(1-\frac{1}{2}(\delta\zeta-\delta\upsilon)^2\right)\right\},\ \left.\left\{x\,z\left(1-\frac{1}{2}(\delta\zeta-\delta\xi)^2\right),\ y\,z\left(1-\frac{1}{2}(\delta\zeta-\delta\upsilon)^2\right),\ z^2\right\}\right\}$$